第 8 章：中断系统

井艳军

沈阳工业大学电气工程学院

主要内容

中断概念和机理

中断相关寄存器

中断响应和处理

中断概念和机理

中断的基本概念

当单片机正在执行程序时，出现了某些特殊状况，例如定时时间到、有键盘信号输入等，此时 CPU 须要暂时停止当前的程序，而转去执行处理这些事件的程序，待执行完这些特定的程序之后，再返回到原先的程序去执行，这就形成了一次“中断”。

“中断”加强了单片机处理突发事件的能力，如果没有中断功能，对可能发生的特殊状况的处理就必须采用定时查询，这样就会浪费大量的 CPU 时间。

因此，中断是单片机中很重要的一个概念，是提高工作效率的重要功能，中断系统功能的好坏是衡量单片机功能的重要指标。

中断源

单片机的“中断源”与单片机包含的外围设备有很大的关系，所谓“中断源”就是引起中断的原因或根源，就是中断请求的来源。

外围设备在工作过程中，都需要 CPU 参与控制、协调或交换数据等服务，而 CPU 正是通过中断技术使得这些外围设备协调工作的。

中断源

中断源种类	中断源标志位	中断源屏蔽位
外部触发中断 INT	INTF	INTE
TMR0 溢出中断	T0IF	T0IE
RB 端口电平变化中断	RBIF	RBIE
TMR1 溢出中断	TMR1IF	TMR1IE
TMR2 中断	TMR2IF	TMR2IE
CCP1 中断	CCP1IF	CCP1IE
CCP2 中断	CCP2IF	CCP2IE
SCI 同步发送中断	TXIF	TXIE
SCI 同步接收中断	RCIF	RCIE
SSP 中断	SSPIF	SSPIE
SSP IIC 总线冲突中断	BCLIF	BCLIE
并行端口中断	PSPIF	PSPIE
A/D 转换中断	ADIF	ADIE
EEPROM 中断	EEIF	EEIE

中断源

从上表可看出，各中断源基本上都与各个外围设备模块相对应的：

- 多数的外围设备对应着一个中断源 (如定时 / 计数器 TMR0)，也有的外围设备对应二个中断源 (如 SCI 同步 / 异步接收 / 发送器 USART)；
- 有的外围设备没有中断源与之对应 (如输入 / 输出端口 RA 和 RC)；
- 也有的中断源没有外围设备与之对应 (例如外部中断源 INT)。

中断处理的过程

- 1）假如 CPU 正在执行更重要的任务，则可采用屏蔽的方法暂时不响应该中断请求；

- 2）如果可以响应该中断请求，则 CPU 执行完当前指令后，必须把断点处的程序计数器 PC 的值 (即下一条指令的地址) 压入堆栈保存起来 (断点保护)，也可以把一些的重要寄存器内容也保护起来 (现场保护)。然后再转移到相应的中断服务子程序中执行。

中断处理的过程

- 3）在中断服务子程序中，首先必须确定发出中断请求的中断源，然后再跳转到与该中断源相对应的程序分支中去执行中断服务程序。

- 4）当中断服务程序执行毕后，必须先恢复被保护的寄存器的值 (现场恢复)，再将程序计数器 PC 的值从堆栈中恢复 (断点恢复) ，使 CPU 返回断点处继续执行被中断的程序。

中断硬件逻辑


EEIF
EEIE
PSPIF
PSPIE
ADIF
ADIE
RCIF
RCIE
TXIF
TXIE
SSPIF
SSPIE
CCPIF
CCPIE
TMR2IF
TMR2IE
TMR1IF
TMR1IE
CCP2IF
CCP2IE
BCLIF
BCLIE
中断源第二梯队
T0IF
T0IE
INTF
INTE
RBIF
RBIE
PEIE
GIE
唤醒CPU(如果
处于睡眠模式)
中断CPU
当前的程序
中断源第一梯队

中断硬件逻辑

在 PIC16F87X 单片机中，不但中断源的种类和个数不同，它们的中断逻辑电路也不同。

每一种中断源对应了一个中断标志位，记为XXXF，以及一个中断屏蔽位或叫中断使能位，记为XXXE。中断源产生的中断信号能否到达 CPU，都受控于相应的中断使能位。

每个中断源申请中断时，其中断标志位会自动置位，中断标志位的清 0 是由用户程序完成的；而每个中断使能位的置位和清位均由用户程序完成。

中断硬件逻辑

图中全部的的 14 个中断源按两个梯队并列排开，第一梯队中只安排了 3 个中断源，其余的中断源全部安排到第二梯队中。

所有的中断源都受“全局中断使能位”(也称总使能位)GIE 的控制；

第一梯队的中断源不仅受 GIE 的控制，还要受各自中断使能位的控制；

第二梯队的中断源不仅受到 GIE 和各自中断使能位的控制，还要受到一个外设中断使能位 PEIE的控制。

中断相关寄存器

中断相关寄存器

与中断有关的特殊功能寄存器 SFR 共有 6 个，分别是：

选项寄存器 OPTION_REG、中断控制寄存器 INTCON、第一外围设备中断标志寄存器 PIR1、第一外围设备中断使能寄存器 PIE1(也称中断使能寄存器)、第二外围设备中断标志寄存器 PIR2 和第二外围设备中断使能寄存器 PIE2，如表所列。

后 5 个 SFR，共有 40 位，但仅使用了 30 位来控制中断，分别与中断逻辑电路的输入信号成严格的对应关系。

中断相关寄存器

bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
OPTION_REG(选项寄存器)							
$\overline{RBPU}$	INTEDG	T0CS	T0SE	PSA	PS2	PS1	PS0
INTCON(中断控制寄存器)							
GIE	PEIE	T0IE	INTE	RBIE	T0IF	INTF	RBIF
PIR1(第一中断标志寄存器)							
PSPIF	ADIF	RCIF	TXIF	SSPIF	CCP1IF	TMR2IF	TMR1IF
PIE1(第一中断使能寄存器)							
PSPIE	ADIE	RCIE	TXIE	SSPIE	CCP1IE	TMR2IE	TMR1IE
PIR2(第二中断标志寄存器)							
–	–	–	EEIF	BCLIF	–	–	CCP2IF
PIE2(第二中断使能寄存器)							
–	–	–	EEIE	BCLIE	–	–	CCP2IE

选项寄存器 OPTION_REG

OPTION_REG(选项寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
$\overline{RBPU}$	INTEDG	T0CS	T0SE	PSA	PS2	PS1	PS0

该寄存器包含了与定时 / 计数器 TMR0、分频器和端口 RB 有关的控制位。RB 端口引脚 RB0 和外部中断 INT 复用一脚，与该脚有关的一个控制位含义如下：

Bit6 / INTEDG：INT 中断信号触发边沿选择位。

清零 0

BR0/INT 引脚上的上升沿触发；

置位 1

BR0/INT 引脚上的下降沿触发。

中断控制寄存器 INTCON

INTCON(中断控制寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
GIE	PEIE	T0IE	INTE	RBIE	T0IF	INTF	RBIF

它将第一梯队中的 3 个中断源的标志位和使能位，以及 PEIE 和 GIE 包含在其中。

中断控制寄存器 INTCON

Bit0 / RBIF：端口 RB 的引脚 RB4～RB7 电平变化中断标志位。

清零 0

RB4～RB7 尚未发生电平变化；

置位 1

B4～RB7 已经发生了电平变化；

Bit3 / RBIE：端口 RB 的引脚 RB4～RB7 电平变化中断使能位。

清零 0

禁止端口 RB 产生的中断

置位 1

允许端口 RB 产生的中断

中断控制寄存器 INTCON

Bit1 / INTF：外部 INT 引脚中断标志位。

清零 0
外部 INT 引脚无中断触发信号

置位 1
外部 INT 引脚有中断触发信号

Bit2 / INTE：外部 INT 引脚中断使能位。

清零 0
禁止外部 INT 引脚产生的中断

置位 1
允许外部 INT 引脚产生的中断

中断控制寄存器 INTCON

Bit2 / T0IF：TMR0 溢出标志位 (也就是溢出中断标志)。

清零 0
TMR0 未发生溢出。

置位 1
TMR0 发生溢出；

Bit5 / T0IE：TMR0 溢出中断使能位。

清零 0
禁止 TMR0 溢出后产生中断。

置位 1
允许 TMR0 溢出后产生中断；

中断控制寄存器 INTCON

Bit6 / PEIE：外设中断使能位。

清零 0
禁止 CPU 响应来自第二梯队的中断请求

置位 1
允许 CPU 响应来自第二梯队的中断请求

中断控制寄存器 INTCON

Bit7 / GIE：全局中断总使能位。

清零 0
禁止 CPU 响应所有外围设备产生的中断请求。

置位 1
允许 CPU 响应所有外围设备产生的中断请求；

第一外围设备中断标志寄存器 PIR1

PIR1(第一中断标志寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
PSPIF	ADIF	RCIF	TXIF	SSPIF	CCP1IF	TMR2IF	TMR1IF

TMR1IF :定时 / 计数器 TMR1 模块溢出中断标志位。

TMR2IF :定时 / 计数器 TMR2 模块溢出中断标志位。

CCP1IF :输入捕捉 / 输出比较 / 脉宽调制模块中断标志位。

SSPIF :同步串行端口 (SSP) 中断标志位。

TXIF :串行通信接口 (SCI) 发送中断标志位。

RCIF :串行通信接口 (SCI) 接收中断标志位。

ADIF :模拟 / 数字 (A / D) 转换中断标志位。

PSPIF :并行端口中断标志位。

第一外围设备中断使能寄存器 PIE1

PIE1(第一中断使能寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
PSPIE	ADIE	RCIE	TXIE	SSPIE	CCP1IE	TMR2IE	TMR1IE

TMR1IE :定时器 / 计数器 TMRI 模块溢出中断使能位。

TMR2IE :定时 / 计数器 TMR2 溢出中断使能位。

CCP1IE :输入捕捉 / 输出比较 / 脉宽调制模块中断使能位。

SSPIE :同步串行端口 (SSP) 中断使能位。

TXIE :串行通信接口 (SCI) 发送中断使能位。

RCIE :串行通信接口 (SCI) 接收中断使能位。

ADIE :模拟 / 数字 (A / D) 转换中断使能位。

PSPIE :并行端口中断使能位。

第二外围设备中断标志寄存器 PIR2

PIR2(第二中断标志寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
–	–	–	EEIF	BCLIF	–	–	CCP2IF

CCP2IF :输入捕捉 / 输出比较 / 脉宽调制 CCP2 模块中断标志位。

BCLIF :I2C 总线冲突中断标志位。

EEIF :EEPROM 写操作中断标志位。

第二外围设备中断使能寄存器 PIE2

PIE2(第二中断使能寄存器)							
bit7	bit6	bit5	bit4	bit3	bit2	bit1	bit0
–	–	–	EEIE	BCLIE	–	–	CCP2IE

CCP2IF :输入捕捉 / 输出比较 / 脉宽调制 CCP2 模块中断使能位。

BCLIF :I2C 总线冲突中断使能位。

EEIF :EEPROM 写操作中断使能位。

中断响应和处理

中断的处理

单片机复位后，硬件会自动设置 GIE=0，屏蔽所有的中断源，假如要求单片机能响应所有中断，必须设置 GIE=1。

其实无论 GIE 或着其他中断屏蔽位是否为 1，当某中断源的中断条件满足时，都会发出中断请求，即相应的中断标志位都会置 1。但是否能够得到 CPU 的响应，则由该中断源的中断使能位决定。

CPU 响应中断后，自动设置 GIE=0，屏蔽所有中断，以免发生重复中断响应，然后自动把 PC 寄存器当前值压入堆栈，并把 PC 值赋予地址0004H，从而转向执行中断服务子程序。

中断的处理

进入中断服务程序后，程序必须要确定发出请求的中断源，这可以通过检查各中断源的标志位来实现。确定了中断源后，就用软件把该中断源的标志位清 0，否则，执行中断返回指令重开中断后，假如中断标志位仍然为 1，则会引起 CPU 重复响应同一个中断请求。

执行中断返回指令 RETFIE 后，不仅会将保留在堆栈中的断点地址弹回到 PC 寄存器中，使程序能返回到被中断的程序处；而且还会自动设置 GIE=1，重新开放所有的中断源。

中断的现场保护问题

中断现场保护是中断技术中十分重要的环节。在进入中断服务子程序期间，堆栈中只能保存返回地址 (程序计数器 PC 的值)。若要保留其他寄存器的内容，怎么办？

由于 PIC 的指令系统中没有 PUSH(入栈) 和 POP(出栈) 之类的指令，所以用户要编制一段程序来实现现场保护的功能。

因为是由用户程序来实现现场保护，而程序的执行中最敏感的寄存器无非就是W 寄存器和STATUS 寄存器，所以应该首先把这两个寄存器保护起来，然后再去保存其他的寄存器。

中断的现场保护问题

PIC 单片机中，中断现场数据是不能保留到堆栈区域中的，而是保留在 RAM 存储器单元中，一般是保留在通用寄存器中。

在 PIC 中，现场保护是很麻烦的事。因为中断是随机发生的，不管程序执行到哪一步，也不管程序所选择的 RAM 的当前体是哪一个，更不管主程序目前在程序存储器那个页面，中断请求及其中断响应都有可能发生。

中断的现场保护问题

在中断现场保护过程中，首先要保存的是累加器 W，这是因为 PIC 单片机，是不可以在不同的文件寄存器之间 (RAM 单元之间) 直接传送数据的，必须要经 W 中转才能实现，所以首先要把 W 寄存器保存，才可以备份其他的寄存器比如 STATUS 。

在保存 W 寄存器时，要确定主程序目前处在的 RAM 存储器的那个体中是十分困难的。虽然可以通过读取 STATUS 寄存器的 RP0 RP1 位得到当前体，但是也是十分麻烦的。

中断的现场保护问题

假设在 RAM 的 bank0 的 20H 上定义一个 W 的临时备份寄存器，通常取名为 W_TEMP。

当利用 MOVWF W_TEMP 命令，企图将 W 中的内容保存到 W_TEMP 中时，往往会出现问题，因为此时保存的内容不一定是在 bank0 中，还和 STATUS 寄存的内容有关系。

为了解决这个问题，PIC16F877 中 RAM 各个体的顶端有 16 个单元，都会彼此映射到相同的 16 个单元上。访问这 16 个单元时体选码就不重要了，因为它们对四个体都是“透明”的。因此，将 W_TEMP 安排在这个部位最为合适，这样做可以使得现场保护和现场恢复变得相对容易实现。

中断的现场保护问题

定义临时变量程序片段：

```
W_TEMP       EQU   70H
STATUS_TEMP  EQU   71H
PCLATH_TEMP  EQU   72H
```

中断的现场保护问题

保护现场程序片段：

```
1 MOVWF    W_TEMP          ; 复制W到备份寄存器W_TEMP中
2 SWAPF    STATUS , W      ; 将STATUS高低半字节交换后放入W
3 MOVWF    STATUS_TEMP     ; 保存STATUS保存到STATUS_TEMP
4 MOVF     PCLATH , W      ; 把寄存器PCLATH内容复制到W中
5 MOVWF    PCLATH_TEMP     ; 将PCLATH保存到PCLATH_TEMP
6 ……                       ; ( 中断服务子程序处理部分 )
```

中断的现场保护问题

回复现场程序片段：

```
1 MOVF     PCLATH_TEMP , W  ; 经过W转移
2 MOVWF    PCLATH           ; 恢复PCLATH内容
3 SWAPF    STATUS_TEMP , W  ; 高低半字节交换后放人W
4 MOVWF    STATUS           ; 把W内容移动到STATUS寄存器
5 SWAPF    W_TEMP , F       ; W_TEMP高低半字节交换放回
6 SWAPF    W_TEMP , W       ; W_TEMP高低半字节交换放入W
```

中断的现场保护程序说明

程序中对要保留的寄存器分别定义了相应的临时备份寄存器，例如“W”的临时备份寄存器为“W_TEMP”。

在保存 W 时，决不能影响状态寄存器 STATUS 的当前内容，否则会影响到体选码及其他标志位。

保存好 W 后，才可以保存 STATUS 和其他寄存器。指令系统总是先将 STATUS 的内容传送 W，再经 W 传送到备份相应的备份寄存器中，比如 STATUS_TEMP。

中断的现场保护程序说明

在程序中使用 SWAPF STATUS,W 来保存状态寄存器 STATUS 的内容，而没有使用 MOVF STATUS,W，因为该指令会影响 STATUS 的 Z 标志位，显然这是不可容忍的。

因此，程序采用字节交换指令 SWAPF 来顶替。在此只利用 SWAPF 的传递功能，它带来的交换功能得记录下来，在返回主程序之前必须把它重新置换回来。

寄存器 PCLATH 的内容保护并不是必须的，只要主程序和中断子程序中都不会修改 PCLATH 的内容，就可以不保护它。

中断的现场恢复程序说明

恢复现场的操作次序与保护现场的操作次序应该相反。

在恢复 STATUS 的内容时候一定要注意把保护现场时采用 SWAPF STATUS，W 产生的多余的交换再交换回来。

最后 2 条指令，将 W_TEMP 内容的高、低半字节交换了 2 遍，才被恢复到工作寄存器 W 中。为不使用 MOVF W_TEMP，W ?

因为 MOVF W_TEMP，W 又会影响 Z 标志位，会破坏刚刚恢复的 STATUS 中的内容，这显然是不可以的。

应用实例

例题 8-1

采用 RB 端口高四位电平变化中断功能，实现从 PORTD（LED）输出手动计数显示，要求点亮发光二极管且具有不断闪烁功能。

程序头

```
1 ;-------------------------------------------------------------
2 LIST     P=16F877
3 INCLUDE "P16F877.INC"
4 ;-------------------------------------------------------------
5 COUNTER    EQU        30H
6 PORTD_TEP  EQU        31H
7            ORG        0000H
8            NOP
9            GOTO       ST
```

中断程序

```
1            ORG       0004H
2            BCF       INTCON, RBIF
3  RB4       BTFSS     PORTB, 4
4            GOTO      RB5
5            CALL      DELAY10MS
6            BTFSS     PORTB, 4
7            GOTO      RB5
8  PP4       BTFSC     PORTB, 4
9            GOTO      PP4
10           CALL      DELAY10MS
11           BTFSC     PORTB, 4
12           GOTO      PP4
13           INCF      PORTD_TEP, F
14           GOTO      RE
```

中断程序

```
1  RB5         BTFSS     PORTB, 5
2              GOTO      RE
3              CALL      DELAY10MS
4              BTFSS     PORTB, 5
5              GOTO      RE
6  PP5         BTFSC     PORTB, 5
7              GOTO      PP5
8              CALL      DELAY10MS
9              BTFSC     PORTB, 5
10             GOTO      PP5
11             DECF      PORTD_TEP, F
12 RE          RETFIE
```

主程序

```
1  ST          BSF       STATUS, RP0
2              MOVLW     00H
3              MOVWF     TRISD
4              MOVLW     30H
5              MOVWF     TRISB
6              BCF       STATUS, RP0
7              MOVLW     88H
8              MOVWF     INTCON
9              CLRF      PORTD
10             CLRF      PORTD_TEP
```

主程序

```
1 LOOP       MOVLW    00H
2            MOVWF    PORTD
3            CALL     DELAY
4            MOVF     PORTD_TEP, W
5            MOVWF    PORTD
6            CALL     DELAY
7            GOTO     LOOP
```

主程序

```
1 DELAY10MS MOVLW    0DH
2           MOVWF    20H
3 LOOP1     MOVLW    0FFH
4           MOVWF    21H
5 LOOP2     DECFSZ   21H
6           GOTO     LOOP2
7           DECFSZ   20H
8           GOTO     LOOP1
9           RETURN
```

主程序

```
1 DELAY     MOVLW   80H
2           MOVWF   20H
3 LP1       MOVLW   0FFH
4           MOVWF   21H
5 LP2       DECFSZ  21H
6           GOTO    LP2
7           DECFSZ  20H
8           GOTO    LP1
9           RETURN
```

小结

除了在中断现场保护部分提到的一些需注意的问题之外，还有前面提到的一些重要注意要注意：

(1) 单片机在任何情况下的复位，均会导致总屏蔽位和其他的中断屏蔽位清 0，即在默认状态下，禁止 CPU 响应所有中断。

(2) 中断标志位的状态与该所有的中断屏蔽位无关，即不管是否允许 CPU 响应中断源的中断请求，只要满足了中断的条件，中断标志位就会被置成 1。

小结

(3) 当系统开放某一中断源时，中断源就通过中断标志位向 CPU 申请中断，只要将中断标志位置 1，就会产生中断响应。

(4) 当 CPU 响应任一个中断后，全局中断屏蔽位 GIE 将会自动清 0；当中断返回时它又会自动恢复为 1。

(5) 如果在中断服务期间若用软件将自动清 0 的 GIE 重新置 1，这时若再出现中断请求，就会形成中断的嵌套：即在响应某一中断期间又响应了其他中断。不过嵌套的级数不能超过堆栈的深度 (8 级)，以免造成堆栈溢出。

小结

(6) 如果同时发生多个中断请求，到底哪个中断会优先得到处理，完全取决于在中断服务子程序中检查中断源的顺序，原因是各个中断源之间不存在优先级别之分。

(7) 每一种中断源受屏蔽的次数不完全相同，第一梯队的中断源受到二次屏蔽，而第二梯队的中断源受到三次屏蔽。

(8) PIC 系列单片机的型号不同，数据存储器 RAM 的布局不完全相同，为工作寄存器 W 安排备份寄存器 W_TEMP 的方法也就不完全相同。